Pensieve Header: Wheeled Semi-Symmetrized calculus in the 2D quotient: Solving for the “conj” coefficients.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\w-Computations"]
```

```
C:\drorbn\AcademicPensieve\Projects\w-Computations
```

```
ar[i_, j_] := t[i] h[j]
```

```
μCollect[μ_] := Collect[μ, _h, Collect[#, _t, FullSimplify] &];
μForm[μ_] := Module[
  {tails, heads, mat},
  tails = Union[Cases[μ, t[s_] :→ s, Infinity]];
  heads = Union[Cases[μ, h[s_] :→ s, Infinity]];
  mat = Outer[Coefficient[μ, h[#1] t[#2]] &, heads, tails];
  PrependTo[mat, t /@ tails];
  mat = Prepend[Transpose[mat],
    Prepend[h /@ heads, μ /. (h[_] | t[_]) → 0]
   ];
  MatrixForm[mat]
 ]
```

```
hm[x_, y_, z_][μ_] := Module[
  {ξ, η},
  ξ = D[μ, h[x]];
  η = D[μ, h[y]];
  μCollect[(μ /. h[x | y] → 0) + ξ h[z] + (1 + ξ /. t[i_] :→ c[i]) η h[z]]
 ]
```

```
hm[3, 4, 5][ar[1, 3] + ar[2, 4]]
```

```
h[5] (t[1] + (1 + c[1]) t[2])
```

```
hm[3, 4, 5][ar[1, 3] + ar[2, 4]] // μForm
```

$$\begin{pmatrix} 0 & h[5] \\ t[1] & 1 \\ t[2] & 1+c[1] \end{pmatrix}$$

```
μ1 = ar[1, 4] + ar[2, 5] + ar[3, 6]
```

```
h[4] t[1] + h[5] t[2] + h[6] t[3]
```

```
μ1 // μForm
```

$$\begin{pmatrix} 0 & h[4] & h[5] & h[6] \\ t[1] & 1 & 0 & 0 \\ t[2] & 0 & 1 & 0 \\ t[3] & 0 & 0 & 1 \end{pmatrix}$$

```
hm[4, 5, 7][μ1]
```

```
h[7] (t[1] + (1 + c[1]) t[2]) + h[6] t[3]
```

```
hm[7, 6, 8][hm[4, 5, 7][μ1]]
```

```
h[8] (t[1] + (1 + c[1]) t[2] + (1 + c[1]) (1 + c[2]) t[3])
```

```
hm[4, 7, 8][hm[5, 6, 7][μ1]] // μForm
```

$$\begin{pmatrix} 0 & h[8] \\ t[1] & 1 \\ t[2] & 1+c[1] \\ t[3] & (1+c[1])(1+c[2]) \end{pmatrix}$$

```
hm[7, 6, 8][hm[4, 5, 7][μ1]] - hm[4, 7, 8][hm[5, 6, 7][μ1]]
```

```
0
```

```
hfac[z_, xtails_List → x_, y_][μ_] := Module[
   {ytails},
   ytails = Complement[
     Union[Cases[μ, t[s_] :→ s, Infinity]],
     xtails
    ];
   hfac[z, xtails → x, ytails → y][μ]
  ];
hfac[z_, x_, ytails_List → y_][μ_] := Module[
   {xtails},
   xtails = Complement[
     Union[Cases[μ, t[s_] :→ s, Infinity]],
     ytails
    ];
   hfac[z, xtails → x, ytails → y][μ]
  ];
hfac[z_, xtails_List → x_, ytails_List → y_][μ_] := Module[
  {ζ, ξ, η},
  ζ = D[μ, h[z]];
  ξ = ζ /. ((t[#] → 0) & /@ ytails);
  η = ζ /. ((t[#] → 0) & /@ xtails);
  μCollect[μ - h[z] ζ + h[x] ξ + h[y] η / (1 + ξ /. t[s_] → c[s])]
 ]
```

```
hm[3, 4, 5][ar[1, 3] + ar[2, 4]] // μForm
```

$$\begin{pmatrix} 0 & h[5] \\ t[1] & 1 \\ t[2] & 1+c[1] \end{pmatrix}$$

```
hm[3, 4, 5][ar[1, 3] + ar[2, 4]] // hfac[5, {1} → 3, 4]
```

```
h[3] t[1] + h[4] t[2]
```

```
(μ2 = α1 ar[1, 1] + α2 ar[2, 1] + α3 ar[2, 2] + α4 ar[2, 3]) // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & 0 & 0 \\ t[2] & \alpha_2 & \alpha_3 & \alpha_4 \end{pmatrix}$$

conj1 is designed to work only in the no-feedback case.

```
conj1[y_, x_][μ_] := Module[
  {ξ, η, a},
  ξ = D[μ, h[x]];
  η = D[μ, t[y]];
  a = a1 + a2 (ξ /. t[s_] ⧴ c[s]);
  μCollect[(μ /. t[y] → a t[y]) + ξ η]
 ];
```

```
μ2 // conj1[1, 2] // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 (a1 + a2\, c[2]\, \alpha_3) & 0 & 0 \\ t[2] & \alpha_2 + \alpha_1 \alpha_3 & \alpha_3 & \alpha_4 \end{pmatrix}$$

```
μ2 // conj1[1, 2] // conj1[1, 3] // hm[2, 3, 2] // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 (a1 + a2\, c[2]\, \alpha_3)(a1 + a2\, c[2]\, \alpha_4) & 0 \\ t[2] & \alpha_2 + \alpha_1 (\alpha_3 + (a1 + a2\, c[2]\, \alpha_3)\, \alpha_4) & \alpha_4 + \alpha_3 (1 + c[2]\, \alpha_4) \end{pmatrix}$$

```
μ2 //  hm[2, 3, 2] // conj1[1, 2] // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 (a1 + a2\, c[2]\, (\alpha_4 + \alpha_3 (1 + c[2]\, \alpha_4))) & 0 \\ t[2] & \alpha_2 + \alpha_1 (\alpha_4 + \alpha_3 (1 + c[2]\, \alpha_4)) & \alpha_4 + \alpha_3 (1 + c[2]\, \alpha_4) \end{pmatrix}$$

```
Solve[{
  α1 (a1 + a2 c[2] α3) (a1 + a2 c[2] α4) == α1 (a1 + a2 c[2] (α4 + α3 (1 + c[2] α4))),
  α2 + α1 (α3 + (a1 + a2 c[2] α3) α4) == α2 + α1 (α4 + α3 (1 + c[2] α4))
 }, {a1, a2}]
```

{{a1 → 1, a2 → 1}}

conj2 is designed to work only in the feedback case.

```
conj2[y_, x_][μ_] := Module[
  {γ},
  γ = Coefficient[μ, ar[y, x]];
  μCollect[μ /. t[y] → t[y] / (1 + 0 a2 γ)]
 ];
```

```
(μ3 = α1 ar[1, 1] + α2 ar[2, 1] + α3 ar[1, 2] + α4 ar[1, 3]) // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_3 & \alpha_4 \\ t[2] & \alpha_2 & 0 & 0 \end{pmatrix}$$

```
μ3 // conj2[1, 2] // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_3 & \alpha_4 \\ t[2] & \alpha_2 & 0 & 0 \end{pmatrix}$$

```
μ3 // conj2[1, 2] // conj2[1, 3] // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_3 & \alpha_4 \\ t[2] & \alpha_2 & 0 & 0 \end{pmatrix}$$

```
μ3 // conj2[1, 2] // conj2[1, 3] // hm[2, 3, 2] // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & \alpha_4 + \alpha_3 (1 + c[1] \alpha_4) \\ t[2] & \alpha_2 & 0 \end{pmatrix}$$

```
μ3 //  hm[2, 3, 2] //  μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & \alpha_4 + \alpha_3 (1 + c[1] \alpha_4) \\ t[2] & \alpha_2 & 0 \end{pmatrix}$$

```
μ3 //  hm[2, 3, 2] // conj2[1, 2] // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & \alpha_4 + \alpha_3 (1 + c[1] \alpha_4) \\ t[2] & \alpha_2 & 0 \end{pmatrix}$$